

Calculs EGMF

propagation 3D des leptons

Propagation 3D des leptons dans un champ magnetique extragalactique
pour le code Monte-Carlo de cascades electromagnetiques

Objet. Ce document presente une formulation autonome pour introduire l'EGMF dans le modele de cascade du memoire. La propagation est ecrite sous forme vectorielle, avec decomposition parallele/perpendiculaire au champ magnetique, rotation de Rodrigues et integration vectorielle de la position. Les commentaires de code sont donnees sous forme de blocs Fortran 95 a adapter aux noms exacts des routines existantes.

Table des matières

1	Objet et decision de conception	3
2	Conventions, variables et unites	3
2.1	Cosmologie	3
2.2	Champ magnetique	3
2.3	Etat d'une particule chargee	4
3	Equations du mouvement	4
3.1	Forme covariante de depart	4
3.2	Forme trois-vectorielle recommandee	4
3.3	Solution sans champ : test obligatoire	5
4	Propagation rapide par rotation vectorielle	5
4.1	Hypotheses d'un segment magnetique	5
4.2	Angle de rotation	5
4.3	Impulsion finale	6
4.4	Position : integrale vectorielle	6
4.5	Rayon de Larmor pour tests locaux	6
5	Choix de formulation et points de controle	7
6	Algorithme recommande	7
6.1	Decoupage d'un trajet	7
6.2	Propagation d'un segment	7
6.3	Quadrature pratique de la position	8
7	Interaction avec le Monte-Carlo du memoire	8
7.1	Tirage du redshift d'interaction	8
7.2	Quand il faut modifier le tirage	9
7.3	Accumulation des interactions Compton	9
8	Extension synchrotron optionnelle	9
9	Commentaires d'implementation dans le code	9
9.1	Parametres EGMF dans Modules.f95	10
9.2	Routine de champ magnetique	10
9.3	Rotation de Rodrigues	11
9.4	Fonctions de base	11
9.5	Propagation d'un lepton : squelette	11
9.6	Propagation dans une cellule de champ constant	12
9.7	Integrale de rotation	13
9.8	Point d'appel dans le programme principal	14
9.9	Sorties utiles pour image et retards	14
10	Gestion des cas limites	14
10.1	Champ nul ou quasi nul	14
10.2	Impulsion presque parallele au champ	14
10.3	Champ aligne avec un axe	14
10.4	Grandes rotations	14
11	Plan de validation	15

12 Implementation progressive conseillée	15
13 Annexe A - variante si l'on veut conserver une base magnetique	16
14 Annexe B - liste de controle des unites	16
15 Conclusion	16

1 Objet et decision de conception

Le modele du memoire simule des cascades electromagnetiques intergalactiques en trois dimensions. Les photons gamma creent des paires e^+/e^- sur les fonds diffus, puis les leptons diffusent surtout les photons du CMB par Compton inverse. L'EGMF agit seulement sur les particules chargees : il courbe les trajectoires, modifie les directions d'emission des photons secondaires, et introduit des effets angulaires et temporels. Pour les champs typiques de l'ordre de 10^{-15} G, les pertes synchrotron sont tres petites devant les pertes Compton, mais la deflexion spatiale et le retard temporel peuvent etre observables.

L'integration se fait en deux niveaux :

1. **implementation par deflexion seule**, recommandee pour une premiere integration au code existant ;
2. **extension synchrotron optionnelle**, a activer seulement apres validation de la propagation magnetique et de la conservation numerique.

Le choix central est le suivant : les composantes de vitesse ne sont pas propagees directement dans une base magnetique locale. On propage le vecteur impulsion physique adimensionne

$$\mathbf{u} \equiv \frac{\mathbf{p}}{m_e c} = \gamma \boldsymbol{\beta}, \quad \gamma = \sqrt{1 + \|\mathbf{u}\|^2}, \quad \boldsymbol{\beta} = \frac{\mathbf{u}}{\gamma}. \quad (1)$$

La force magnetique fait tourner $\mathbf{u}$ sans changer sa norme instantanee. L'expansion cosmologique multiplie sa norme par $(1+z)$, comme dans le transport sans champ magnetique.

Choix pratique. La routine de propagation des leptons peut etre organisee autour d'une routine vectorielle : decomposition parallele/perpendiculaire a $\mathbf{B}_{\text{phys}}$, rotation de Rodrigues autour de $\mathbf{B}_{\text{phys}}$, puis integration vectorielle de la position. Cette formulation ne depend d'aucune base locale particuliere et se controle facilement par des tests unitaires.

2 Conventions, variables et unites

2.1 Cosmologie

On garde le modele plat Λ CDM utilise dans le memoire :

$$H(z) = H_0 E(z), \quad E(z) = \sqrt{\Omega_M (1+z)^3 + \Omega_\Lambda}, \quad \Omega_k = 0. \quad (2)$$

Sauf mention contraire, on pose $a_0 = 1$, donc

$$a(z) = \frac{1}{1+z}, \quad \frac{dt}{dz} = -\frac{1}{(1+z)H(z)}. \quad (3)$$

Les coordonnees $\mathbf{x} = (x, y, z)$ du code sont prises comme des coordonnees comobiles, en cm si le reste du code est en CGS. La position physique est $\mathbf{r}_{\text{phys}} = a(t)\mathbf{x}$.

2.2 Champ magnetique

Toutes les equations ci-dessous sont en systeme gaussien/CGS. Le champ passe a la force de Lorentz sous la forme physique mesuree par un observateur comobile local :

$$\mathbf{B}_{\text{phys}} = B_x \mathbf{e}_x + B_y \mathbf{e}_y + B_z \mathbf{e}_z, \quad [B] = \text{G}. \quad (4)$$

Il faut choisir une convention unique dans le code :

— si la fonction de champ retourne directement $\mathbf{B}_{\text{phys}}(z)$, on l'utilise sans autre facteur ;

— si l'utilisateur fournit un champ comobile constant $\mathbf{B}_{\text{com}}$, alors

$$\mathbf{B}_{\text{phys}}(z) = \frac{\mathbf{B}_{\text{com}}}{a(z)^2} = \mathbf{B}_{\text{com}}(1+z)^2. \quad (5)$$

Point a ne pas laisser implicite. Un nombre tel que $B_0 = 10^{-15} \text{ G}$ peut designer soit le champ physique aujourd'hui, soit un champ physique constant dans chaque cellule, soit un champ comobile. Ces trois choix ne donnent pas la meme rotation a grand redshift. Le code doit contenir un seul endroit ou le facteur $(1+z)^2$ est eventuellement applique.

2.3 Etat d'une particule chargee

Pour un lepton, on stocke au minimum :

$$(\mathbf{x}, z, q, \gamma, \hat{\mathbf{n}}, w), \quad q = \begin{cases} -e & e^- \\ +e & e^+ \end{cases} \quad (6)$$

ou w est le poids statistique Monte-Carlo. La direction physique est $\hat{\mathbf{n}}$ et

$$\mathbf{u} = \sqrt{\gamma^2 - 1} \hat{\mathbf{n}}. \quad (7)$$

Apres propagation, on reconstruit

$$\gamma_f = \sqrt{1 + \|\mathbf{u}_f\|^2}, \quad \hat{\mathbf{n}}_f = \frac{\mathbf{u}_f}{\|\mathbf{u}_f\|}. \quad (8)$$

3 Equations du mouvement

3.1 Forme covariante de depart

La forme covariante compacte est

$$\frac{DU^\mu}{d\tau} = \frac{q}{m_e} F^\mu{}_\nu U^\nu, \quad (9)$$

ou $U^\mu = dx^\mu/d\tau$ est la quadrivitesse. Cette formule est utile pour verifier le formalisme. Pour le code, on utilisera la forme trois-vectorielle ci-dessous. Dans un univers de Robertson-Walker plat,

$$ds^2 = c^2 dt^2 - a^2(t) d\mathbf{x}^2, \quad (10)$$

on obtient une forme trois-vectorielle beaucoup plus lisible en impulsion physique.

3.2 Forme trois-vectorielle recommandee

Avec $\mathbf{u} = \mathbf{p}/(m_e c)$, les equations a integrer sont

$$\frac{d\mathbf{x}}{dt} = \frac{c}{a(t)} \frac{\mathbf{u}}{\gamma}, \quad (11)$$

$$\frac{d\mathbf{u}}{dt} = -H(z)\mathbf{u} + \frac{q}{m_e c \gamma} \mathbf{u} \times \mathbf{B}_{\text{phys}}, \quad (12)$$

$$\frac{dz}{dt} = -(1+z)H(z). \quad (13)$$

Le terme $-H(z)\mathbf{u}$ est le redshift cosmologique de l'impulsion physique. Le terme magnetique est une rotation pure autour de $\mathbf{B}_{\text{phys}}$:

$$\left. \frac{d}{dt} \|\mathbf{u}\|^2 \right|_B = 2\mathbf{u} \cdot \frac{q}{m_e c \gamma} (\mathbf{u} \times \mathbf{B}_{\text{phys}}) = 0. \quad (14)$$

Le champ magnetique seul ne change donc pas l'energie.

En utilisant z comme variable d'integration, on a aussi

$$\frac{d\mathbf{x}}{dz} = -\frac{c}{H_0} \frac{\mathbf{u}}{\gamma E(z)}, \quad (15)$$

$$\frac{d\mathbf{u}}{dz} = \frac{\mathbf{u}}{1+z} - \frac{q}{m_e c H_0} \frac{\mathbf{u} \times \mathbf{B}_{\text{phys}}}{(1+z)\gamma E(z)}. \quad (16)$$

Ces deux equations sont une solution de repli simple si le champ varie en direction a l'interieur d'un pas. Pour une cellule de champ constant, la methode de rotation de la section suivante est plus robuste.

3.3 Solution sans champ : test obligatoire

Si $\mathbf{B}_{\text{phys}} = 0$, la norme de l'impulsion adimensionnee verifie

$$\mathbf{u}(z) = \mathbf{u}_i \frac{1+z}{1+z_i}. \quad (17)$$

Donc

$$\gamma(z) = \sqrt{1 + (\gamma_i^2 - 1) \left(\frac{1+z}{1+z_i} \right)^2}. \quad (18)$$

Cette expression est a conserver : elle est identique avec ou sans champ magnetique tant que l'on ignore le synchrotron.

La distance comobile vectorielle parcourue sans champ entre z_i et $z_f < z_i$ est

$$\Delta \mathbf{x} = \frac{c}{H_0} \int_{z_f}^{z_i} \frac{\mathbf{u}(z)}{\gamma(z) E(z)} dz. \quad (19)$$

Si la direction reste constante, cela se reduit a l'integrale scalaire de parcours du memoire, a un facteur a_0 pres si le code ne pose pas $a_0 = 1$.

4 Propagation rapide par rotation vectorielle

4.1 Hypotheses d'un segment magnetique

On decoupe la trajectoire en segments tels que, sur chaque segment :

- la direction du champ $\hat{\mathbf{b}} = \mathbf{B}_{\text{phys}} / \|\mathbf{B}_{\text{phys}}\|$ est constante ;
- la convention d'amplitude $B_{\text{phys}}(z)$ est connue ;
- aucun evenement Monte-Carlo n'a lieu a l'interieur du segment ;
- les pertes synchrotron sont ignorees dans la configuration de base.

On decompose l'impulsion initiale :

$$\mathbf{u}_{\parallel,i} = (\mathbf{u}_i \cdot \hat{\mathbf{b}}) \hat{\mathbf{b}}, \quad (20)$$

$$\mathbf{u}_{\perp,i} = \mathbf{u}_i - \mathbf{u}_{\parallel,i}. \quad (21)$$

4.2 Angle de rotation

Pour $\mathbf{B}_{\text{phys}} = B\hat{\mathbf{b}}$, la vitesse angulaire cyclotron relativiste est

$$\omega_B(z) = \frac{q B_{\text{phys}}(z)}{m_e c \gamma(z)}. \quad (22)$$

Avec l'orientation standard de la rotation de Rodrigues autour de $\hat{\mathbf{b}}$, l'angle accumule entre z_i et z_f est

$$\phi_f = -\frac{q}{m_e c H_0} \int_{z_f}^{z_i} \frac{B_{\text{phys}}(z)}{(1+z)\gamma(z)E(z)} dz. \quad (23)$$

Le signe de q doit etre conserve : electron et positron tournent en sens opposes.

La duree propre correspondante est

$$\Delta\tau = \frac{1}{H_0} \int_{z_f}^{z_i} \frac{dz}{(1+z)\gamma(z)E(z)}, \quad (24)$$

et, si B_{phys} est constant sur le segment,

$$\phi_f = -\frac{qB}{m_e c} \Delta\tau. \quad (25)$$

4.3 Impulsion finale

La rotation de Rodrigues d'un vecteur $\mathbf{v}$ autour d'un axe unitaire $\hat{\mathbf{b}}$ est

$$R_{\hat{\mathbf{b}}}(\phi)\mathbf{v} = \mathbf{v} \cos \phi + (\hat{\mathbf{b}} \times \mathbf{v}) \sin \phi + \hat{\mathbf{b}}(\hat{\mathbf{b}} \cdot \mathbf{v})(1 - \cos \phi). \quad (26)$$

Comme $\mathbf{u}_{\perp,i} \cdot \hat{\mathbf{b}} = 0$, le dernier terme est nul pour la partie perpendiculaire, mais il est utile de garder la forme generale dans le code.

La solution d'un segment est

$$\mathbf{u}(z) = S(z) \left[\mathbf{u}_{\parallel,i} + R_{\hat{\mathbf{b}}}(\phi(z))\mathbf{u}_{\perp,i} \right], \quad S(z) = \frac{1+z}{1+z_i}. \quad (27)$$

Au point final :

$$\mathbf{u}_f = S_f \left[\mathbf{u}_{\parallel,i} + R_{\hat{\mathbf{b}}}(\phi_f)\mathbf{u}_{\perp,i} \right], \quad S_f = \frac{1+z_f}{1+z_i}. \quad (28)$$

Avantage numerique. Cette methode preserve exactement la norme magnetique de $\mathbf{u}$ dans un segment et ne contient aucun angle θ_i calcule par des cas particuliers. Elle fonctionne pour $B_{\text{phys}} \parallel \mathbf{e}_x$, $B_{\text{phys}} \parallel \mathbf{e}_y$, $B_{\text{phys}} \parallel \mathbf{e}_z$ et pour toute direction aleatoire.

4.4 Position : integrale vectorielle

La position comobile finale est

$$\mathbf{x}_f - \mathbf{x}_i = \frac{c}{H_0} \int_{z_f}^{z_i} \frac{\mathbf{u}(z)}{\gamma(z)E(z)} dz. \quad (29)$$

Cette expression est la grandeur de production pour les positions 3D, les images et les retards. Le rayon de Larmor de la section suivante sert uniquement aux comparaisons locales en espace plat et aux tests de petit angle.

4.5 Rayon de Larmor pour tests locaux

Le rayon de Larmor physique relativiste est

$$R_L = \frac{p_{\perp} c}{|q|B} = \frac{\gamma m_e v_{\perp} c}{|q|B} = \frac{m_e c^2 u_{\perp}}{|q|B}. \quad (30)$$

Il sert aux tests unitaires locaux. La position cosmologique reste calculee avec l'integrale vectorielle.

Pour de petits angles sur une distance physique d ,

$$\delta\theta \simeq \frac{d}{R_L} = \frac{|q|Bd}{p_{\perp} c}, \quad (31)$$

ce qui permet de retrouver l'ordre de grandeur de la formule de deflexion donnee dans le memoire.

5 Choix de formulation et points de controle

Sujet	Formulation retenue	Controle numerique
Base magnetique locale	Utiliser directement l'axe unitaire $\hat{\mathbf{b}} = \mathbf{B}_{\text{phys}} / \ \mathbf{B}_{\text{phys}}\ $ et la rotation de Rodrigues. Une base locale n'est pas necessaire pour obtenir les composantes parallele et perpendiculaire.	Tester $\mathbf{B}_{\text{phys}}$ aligne avec chacun des trois axes globaux et avec une direction aleatoire.
Composantes parallele et perpendiculaire	Decomposer $\mathbf{u}_{\parallel} = (\mathbf{u} \cdot \hat{\mathbf{b}})\hat{\mathbf{b}}$ et $\mathbf{u}_{\perp} = \mathbf{u} - \mathbf{u}_{\parallel}$.	Verifier $\mathbf{u}_{\perp} \cdot \hat{\mathbf{b}} = 0$ a la precision machine.
Variable dynamique	Propager $\mathbf{u} = \gamma\beta$ plutot que les composantes de vitesse. L'expansion donne simplement $\mathbf{u}(z) = \mathbf{u}_i(1+z)/(1+z_i)$ hors rotation.	Verifier $\ \mathbf{u}_f\ / \ \mathbf{u}_i\ = (1+z_f)/(1+z_i)$ sans synchrotron.
Rayon de Larmor	Utiliser $R_L = p_{\perp}c/(q B)$ comme longueur physique locale de reference.	Comparer la deflexion de petit angle a $\delta\theta \simeq d/R_L$.
Position 3D	Calculer $\Delta\mathbf{x}$ par l'integrale vectorielle (29).	Reduire le pas en angle ϕ_{max} et verifier la convergence de l'image et des retards.
Convention de champ $B(z)$	Centraliser la convention dans EGMF_Bphys : champ physique local ou champ comobile converti en champ physique.	Comparer explicitement les cas B_mode=0 et B_mode=1 sur un test court a redshift fixe.

6 Algorithme recommande

6.1 Decoupage d'un trajet

Un lepton apparait a un redshift z_i et doit etre propage jusqu'au prochain redshift d'interaction z_f tire par Monte-Carlo, ou jusqu'a une condition limite. On decoupe $[z_f, z_i]$ en sous-segments dont les bornes sont donnees par :

- le redshift final du trajet demande ;
- les frontieres de cellule magnetique, si le champ est coherent par cellules ;
- un critere de resolution, par exemple $|\Delta\phi| < \phi_{\text{max}}$ avec $\phi_{\text{max}} \simeq 0.05$ a 0.1 rad ;
- eventuellement une limite relative sur $\Delta\gamma/\gamma$ pour les tests non ultra-relativistes.

6.2 Propagation d'un segment

Pour chaque segment $z_{\text{old}} \rightarrow z_{\text{new}}$ avec $z_{\text{new}} < z_{\text{old}}$:

1. evaluer $\mathbf{B}_{\text{phys}}$ a la position courante, avec la bonne convention de redshift ;
2. si $\|\mathbf{B}_{\text{phys}}\| < B_{\text{floor}}$, utiliser la propagation sans champ ;
3. sinon calculer $\hat{\mathbf{b}}, \mathbf{u}_{\parallel}, \mathbf{u}_{\perp}$;
4. integrer l'angle (23) ;

5. integrer la position avec (29), en utilisant (27) dans l'integrande ;
6. mettre a jour $\mathbf{u}$, γ , la direction, la position et le redshift.

6.3 Quadrature pratique de la position

La facon la plus simple consiste a reutiliser la quadrature de Gauss-Legendre deja presente dans le code. Pour chaque point de quadrature z_j :

$$\gamma_j = \sqrt{1 + (\gamma_i^2 - 1) \left(\frac{1 + z_j}{1 + z_i} \right)^2}, \quad (32)$$

$$\phi_j = -\frac{q}{m_e c H_0} \int_{z_j}^{z_i} \frac{B_{\text{phys}}(z')}{(1 + z')\gamma(z')E(z')} dz', \quad (33)$$

$$\mathbf{u}_j = \frac{1 + z_j}{1 + z_i} [\mathbf{u}_{\parallel, i} + R_{\hat{\mathbf{b}}}(\phi_j)\mathbf{u}_{\perp, i}], \quad (34)$$

$$\mathbf{I}_j = \frac{\mathbf{u}_j}{\gamma_j E(z_j)}. \quad (35)$$

Puis

$$\Delta \mathbf{x} = \frac{c}{H_0} \int_{z_f}^{z_i} \mathbf{I}(z) dz. \quad (36)$$

Pour eviter une integrale imbriquee dans ϕ_j , on peut aussi integrer simultanement ϕ et $\mathbf{x}$ avec la variable positive $s = z_i - z$:

$$\frac{d\phi}{ds} = -\frac{q}{m_e c H_0} \frac{B_{\text{phys}}(z_i - s)}{(1 + z_i - s)\gamma(s)E(z_i - s)}, \quad (37)$$

$$\frac{d\mathbf{x}}{ds} = \frac{c}{H_0} \frac{\mathbf{u}(s)}{\gamma(s)E(z_i - s)}. \quad (38)$$

Ici $\gamma(s)$ et $\mathbf{u}(s)$ designent respectivement $\gamma(z_i - s)$ et $\mathbf{u}(z_i - s)$. Un schema RK4 ou Gauss-Legendre vectoriel suffit, car l'integrande est lisse tant que le champ est constant dans la cellule.

7 Interaction avec le Monte-Carlo du memoire

7.1 Tirage du redshift d'interaction

Sans synchrotron, l'EGMF ne change pas $\gamma(z)$, seulement la direction et la position. Si les fonds CMB/EBL sont isotropes et homogenes a redshift fixe, le tirage du redshift d'interaction peut rester celui du code existant. La profondeur optique Compton cumulee, notee ici τ_c , s'ecrit

$$\tau_c = \frac{c}{H_0} \int_{z_f}^{z_i} \frac{(d\tau_c/dx)(E_e(z), z) \beta(z)}{(1 + z)E(z)} dz, \quad (39)$$

avec

$$\beta(z) = \frac{\sqrt{\gamma(z)^2 - 1}}{\gamma(z)}. \quad (40)$$

Dans cette notation, $(d\tau_c/dx)$ designe le taux d'interaction par unite de longueur physique, sans facteur de vitesse deja inclus. Si une routine existante stocke un taux par unite de longueur comobile, ou une grandeur qui inclut deja β , la conversion doit etre centralisee dans un seul wrapper avant l'integration. La courbure magnetique n'apparait pas dans cette integrale si les champs de photons sont isotropes : l'interaction depend de l'energie et du redshift, pas de l'orientation absolue.

7.2 Quand il faut modifier le tirage

Il faut recalculer le hazard Monte-Carlo si l'une des conditions suivantes est vraie :

- pertes synchrotron activees, donc $\gamma(z)$ n'est plus donne par (18) ;
- fond de photons anisotrope ou inhomogene ;
- champ magnetique couple a une structure spatiale qui modifie aussi la densite de photons ou la source ;
- acceleration Compton qui regroupe des photons emis a des positions differentes, alors que l'on etudie les images ou les retards.

7.3 Accumulation des interactions Compton

L'accumulation Compton du memoire est utile pour les spectres, mais elle peut biaiser une etude EGMF. En effet, plusieurs photons secondaires devraient etre emis en differents points d'une trajectoire courbe ; les regrouper en un seul evenement deforme l'image et les temps d'arrivee.

Recommandation. Pour les simulations d'image ou de retard temporel avec EGMF, introduire une option `UseAccComptonForEGMF=.false..` Si l'accumulation reste active, distribuer les photons accumules le long du segment propage au lieu de les creer tous au meme point.

8 Extension synchrotron optionnelle

Pour le champ $B \sim 10^{-15}$ G discute dans le memoire, les pertes synchrotron sont negligeables devant les pertes Compton. Elles peuvent donc etre desactivees dans la premiere etape. Si l'on souhaite les ajouter, la puissance synchrotron instantanee avec dependance en angle de pitch est

$$P_{\text{syn}} = 2\sigma_T c U_B \gamma^2 \beta_{\perp}^2, \quad U_B = \frac{B^2}{8\pi}, \quad \beta_{\perp} = \frac{\|\mathbf{u}_{\perp}\|}{\gamma} = \beta \sin \alpha. \quad (41)$$

Pour une distribution isotrope des angles de pitch, la moyenne equivalente est

$$\langle P_{\text{syn}} \rangle = \frac{4}{3} \sigma_T c U_B \gamma^2 \beta^2. \quad (42)$$

L'equation d'energie devient

$$\frac{d\gamma}{dt} = -H(z) \frac{\gamma^2 - 1}{\gamma} - \frac{P_{\text{syn}}}{m_e c^2}. \quad (43)$$

Dans ce cas, l'expression analytique (18) n'est plus valide, et il faut integrer simultanement γ , la direction magnetique et la position. Il faut aussi recalculer la profondeur optique Compton avec cette energie couplee.

Priorite de developpement. Activer d'abord une configuration EGMF sans synchrotron, valider les halos et les retards, puis seulement ajouter le synchrotron continu. L'emission explicite de photons synchrotron peut rester une option separee.

9 Commentaires d'implementation dans le code

Les noms exacts ci-dessous sont indicatifs. Ils reprennent l'organisation decrite dans le memoire : programme principal dans `Cascades.f95`, variables globales dans `Modules.f95`, integrandes dans `Integrand.f95`, routines generales dans `RoutinesFonctions.f95`.

9.1 Parametres EGMF dans Modules.f95

```

1  ! =====
2  ! [EGMF-00] Parametres globaux de propagation magnetique
3  ! =====
4  logical, parameter :: UseEGMF_default = .true.
5  logical             :: UseEGMF
6  logical             :: UseSynchrotron
7  logical             :: UseAccComptonForEGMF
8
9  ! B_mode = 0 : la routine retourne deja B_phys(z) en gauss
10 ! B_mode = 1 : l'utilisateur fournit un champ comobile B_com,
11 !             donc B_phys(z) = B_com * (1.d0 + z)**2
12 integer :: B_mode
13
14 real(dp) :: B_floor_G      ! ex. 1.d-40 : traite comme B=0
15 real(dp) :: phi_max        ! ex. 5.d-2 rad pour decoupage adaptatif
16 real(dp) :: Lcoh_cm        ! longueur de coherence comobile si cellules
17
18 ! [UNITE] H0 doit etre converti en s^-1 avant toute propagation EGMF.
19 ! Exemple : H0_s = H0_km_s_Mpc * 1.d5 / Mpc_cm
20 real(dp) :: H0_s
21
22 ! Constantes CGS si elles ne sont pas deja presentes
23 real(dp), parameter :: c_cgs = 2.99792458d10
24 real(dp), parameter :: me_cgs = 9.1093837d-28
25 real(dp), parameter :: qe_cgs = 4.8032047d-10
26 real(dp), parameter :: sigmaT_cgs = 6.6524587d-25

```

9.2 Routine de champ magnetique

```

1  subroutine EGMF_Bphys(x_comov, zred, Bphys)
2  ! [EGMF-01] Retourne le champ physique local en gauss.
3  ! x_comov : position comobile en cm
4  ! zred    : redshift cosmologique
5  ! Bphys   : champ physique en gauss, dans la base globale du code
6
7  real(dp), intent(in)  :: x_comov(3), zred
8  real(dp), intent(out) :: Bphys(3)
9  real(dp) :: Braw(3)
10
11 ! TODO: adapter ce bloc au modele souhaite :
12 !       champ uniforme, cellules aleatoires, grille 3D, etc.
13 call EGMF_raw_field(x_comov, zred, Braw)
14
15 if (B_mode == 0) then
16   Bphys = Braw
17 else if (B_mode == 1) then
18   Bphys = Braw * (1.d0 + zred)**2
19 else
20   stop 'EGMF_Bphys: B_mode inconnu'
21 end if
22
23 ! [ATTENTION] Ne pas appliquer le facteur (1+z)^2 ailleurs.
24 end subroutine EGMF_Bphys

```

9.3 Rotation de Rodrigues

```

1 subroutine rotate_rodrigues(u, bhat, phi, urot)
2   ! [EGMF-02] Rotation vectorielle robuste autour de bhat.
3   ! bhat doit etre normalise avant l'appel.
4   real(dp), intent(in)  :: u(3), bhat(3), phi
5   real(dp), intent(out) :: urot(3)
6   real(dp) :: cph, sph, bdotu
7   real(dp) :: bxu(3)
8
9   cph = cos(phi)
10  sph = sin(phi)
11  bdotu = dot_product(bhat, u)
12
13  bxu(1) = bhat(2)*u(3) - bhat(3)*u(2)
14  bxu(2) = bhat(3)*u(1) - bhat(1)*u(3)
15  bxu(3) = bhat(1)*u(2) - bhat(2)*u(1)
16
17  urot = u*cph + bxu*sph + bhat*bdotu*(1.d0-cph)
18 end subroutine rotate_rodrigues

```

9.4 Fonctions de base

```

1 function Ez_lcdm(zred) result(Ez)
2   real(dp), intent(in) :: zred
3   real(dp) :: Ez
4   Ez = sqrt(OmegaM*(1.d0+zred)**3 + OmegaLambda)
5 end function Ez_lcdm
6
7 function gamma_no_synch(gamma_i, zred, z_i) result(gam)
8   ! [EGMF-03] Energie si B ne fait que deflechir.
9   real(dp), intent(in) :: gamma_i, zred, z_i
10  real(dp) :: gam, scale, ui2
11  scale = (1.d0 + zred)/(1.d0 + z_i)
12  ui2 = max(gamma_i*gamma_i - 1.d0, 0.d0)
13  gam = sqrt(1.d0 + ui2*scale*scale)
14 end function gamma_no_synch
15
16 subroutine normalize3(v, vhat, vnorm)
17   real(dp), intent(in)  :: v(3)
18   real(dp), intent(out) :: vhat(3), vnorm
19   vnorm = sqrt(dot_product(v,v))
20   if (vnorm > 0.d0) then
21     vhat = v/vnorm
22   else
23     vhat = 0.d0
24   end if
25 end subroutine normalize3

```

9.5 Propagation d'un lepton : squelette

```

1 subroutine propagate_lepton_egmf(x, z_old, z_new, charge_sign, gamma, dir)
2   ! [EGMF-04] Propagation entre deux redshifts sans interaction.
3   ! z_new < z_old.
4   ! charge_sign = -1 pour electron, +1 pour positron.
5   ! x      : position comobile, mise a jour
6   ! gamma  : facteur de Lorentz, mis a jour

```

```

7  ! dir    : direction physique unitaire, mise a jour
8
9  real(dp), intent(inout) :: x(3), gamma, dir(3)
10 real(dp), intent(in)    :: z_old, z_new
11 integer,  intent(in)    :: charge_sign
12
13 real(dp) :: z_a, z_b
14 real(dp) :: u(3), unorm
15
16 if (z_new > z_old) stop 'propagate_lepton_egmf: z_new doit etre < z_old'
17
18 unorm = sqrt(max(gamma*gamma - 1.d0, 0.d0))
19 u = unorm * dir
20
21 z_a = z_old
22 do while (z_a > z_new)
23   ! [EGMF-05] Choisir z_b : minimum entre z_new, frontiere cellule,
24   !               et critere |Delta phi| < phi_max.
25   call choose_next_egmf_step(x, z_a, z_new, gamma, charge_sign, z_b)
26
27   call propagate_one_B_cell(x, z_a, z_b, charge_sign, u)
28
29   gamma = sqrt(1.d0 + dot_product(u,u))
30   if (gamma > 1.d0) then
31     dir = u / sqrt(dot_product(u,u))
32   else
33     ! Cas non relativiste limite : garder la direction courante.
34     dir = dir / sqrt(max(dot_product(dir,dir), tiny(1.d0)))
35   end if
36
37   z_a = z_b
38 end do
39 end subroutine propagate_lepton_egmf

```

9.6 Propagation dans une cellule de champ constant

```

1  subroutine propagate_one_B_cell(x, z_i, z_f, charge_sign, u_i)
2  ! [EGMF-06] Propagation : rotation de u + integrale vectorielle de x.
3  ! Entree/sortie : x et u_i sont mis a jour au redshift z_f.
4
5  real(dp), intent(inout) :: x(3), u_i(3)
6  real(dp), intent(in)    :: z_i, z_f
7  integer,  intent(in)    :: charge_sign
8
9  real(dp) :: B(3), Bnorm, bhat(3)
10 real(dp) :: gamma_i, scale_f, phi_f
11 real(dp) :: upar(3), uperp(3), uperp_rot(3)
12 real(dp) :: dx(3)
13
14 call EGMF_Bphys(x, z_i, B)
15 call normalize3(B, bhat, Bnorm)
16
17 gamma_i = sqrt(1.d0 + dot_product(u_i,u_i))
18
19 if (Bnorm < B_floor_G) then
20   call propagate_no_B(x, z_i, z_f, u_i)
21   return
22 end if

```

```

23
24 upar = dot_product(u_i, bhat) * bhat
25 uperp = u_i - upar
26
27 ! [EGMF-07] Angle signe. qe_cgs est positif ; charge_sign porte le signe.
28 call integrate_phi(z_i, z_f, gamma_i, Bnorm, charge_sign, phi_f)
29
30 ! [EGMF-08] Position : integrer le vecteur u(z)/(gamma E(z)).
31 ! La position vient de la quadrature vectorielle du segment.
32 call integrate_position_segment(z_i, z_f, gamma_i, upar, uperp, &
33                                bhat, Bnorm, charge_sign, dx)
34
35 x = x + dx
36
37 call rotate_rodrigues(uperp, bhat, phi_f, uperp_rot)
38 scale_f = (1.d0 + z_f)/(1.d0 + z_i)
39 u_i = scale_f * (upar + uperp_rot)
40 end subroutine propagate_one_B_cell

```

9.7 Integrale de rotation

```

1 function Bnorm_cell_at_z(Bnorm_ref, zred, z_ref) result(Bnorm_z)
2   ! [EGMF-09a] Amplitude physique dans une cellule de direction fixe.
3   ! Bnorm_ref est l'amplitude physique au redshift z_ref.
4   real(dp), intent(in) :: Bnorm_ref, zred, z_ref
5   real(dp) :: Bnorm_z
6
7   if (B_mode == 0) then
8     ! La routine de champ fournit deja le champ physique local.
9     ! Dans un segment court, on peut le prendre constant ou
10    ! rappeler EGMF_Bphys dans l'integrande si le modele spatial l'exige.
11    Bnorm_z = Bnorm_ref
12  else if (B_mode == 1) then
13    ! B_com constant dans la cellule : B_phys propto (1+z)^2.
14    Bnorm_z = Bnorm_ref * ((1.d0 + zred)/(1.d0 + z_ref))**2
15  else
16    stop 'Bnorm_cell_at_z: B_mode inconnu'
17  end if
18 end function Bnorm_cell_at_z
19
20 subroutine integrate_phi(z_i, z_f, gamma_i, Bnorm_i, charge_sign, phi)
21   ! [EGMF-09b] Calcule phi = -q/(m c H0) * int_zf^zi B/((1+z) gamma E) dz.
22   ! Hypothese : direction de B constante dans la cellule.
23
24   real(dp), intent(in) :: z_i, z_f, gamma_i, Bnorm_i
25   integer, intent(in) :: charge_sign
26   real(dp), intent(out) :: phi
27   real(dp) :: integ
28
29   ! L'integrande utilise :
30   ! gamma_no_synch(gamma_i, z, z_i)
31   ! Bnorm_cell_at_z(Bnorm_i, z, z_i)
32   ! Ez_lcdm(z)
33   call gauss_integral_phi(z_f, z_i, gamma_i, z_i, Bnorm_i, integ)
34
35   phi = -dble(charge_sign) * qe_cgs/(me_cgs*c_cgs*H0_s) * integ
36 end subroutine integrate_phi

```

9.8 Point d'appel dans le programme principal

```

1 ! =====
2 ! [EGMF-10] Dans la boucle Monte-Carlo, juste apres avoir tire z_int
3 !           et avant de calculer l'interaction au point final :
4 ! =====
5 if (particle_type == ELECTRON .or. particle_type == POSITRON) then
6     if (UseEGMF) then
7         call propagate_lepton_egmf(pos, z_current, z_int, charge_sign, gamma, dir)
8     else
9         call propagate_lepton_no_B(pos, z_current, z_int, gamma, dir)
10    end if
11 else if (particle_type == PHOTON) then
12     call propagate_photon(pos, z_current, z_int, dir)
13 end if
14
15 ! [EGMF-11] L'interaction Compton ou gamma-gamma doit utiliser pos, gamma,
16 !           dir et z_int APRES propagation.

```

9.9 Sorties utiles pour image et retards

```

1 ! [EGMF-12] Colonnes recommandees dans Result ou fichier equivalent :
2 ! type, poids, energie, z_creation, z_detection, x, y, z, dirx, diry, dirz,
3 ! t_cosmic_or_lookback, event_id, parent_id
4 !
5 ! [ATTENTION] Eviter d'utiliser le meme nom zf pour le redshift final et
6 !           pour la coordonnee spatiale z. Preferer z_red_f et x(3).

```

10 Gestion des cas limites

10.1 Champ nul ou quasi nul

Si $B < B_{\text{floor}}$, appeler directement la routine sans champ. Cela evite les divisions par B et les instabilites de normalisation.

10.2 Impulsion presque parallele au champ

Si $\|\mathbf{u}_\perp\| / \|\mathbf{u}\| < \epsilon$, la rotation ne change presque rien. La formule de Rodrigues reste valide; il n'est pas necessaire de traiter un angle θ special.

10.3 Champ aligne avec un axe

Aucun cas particulier n'est requis avec Rodrigues. Le test $\mathbf{B}_{\text{phys}} \parallel \mathbf{e}_x$ doit donner le meme niveau de precision que les tests $\mathbf{B}_{\text{phys}} \parallel \mathbf{e}_y$, $\mathbf{B}_{\text{phys}} \parallel \mathbf{e}_z$ et direction aleatoire.

10.4 Grandes rotations

Si $|\phi| \gg 1$ dans un segment, l'integrale de position devient oscillante. Il faut decouper le segment jusqu'a $|\Delta\phi| < \phi_{\text{max}}$. Pour les EGMF faibles typiques, cette contrainte est souvent moins restrictive que les interactions Compton, mais elle protege les simulations a champ fort.

11 Plan de validation

1. **Test $B = 0$.** La routine doit reproduire les spectres, positions et energies de la configuration sans EGMF.
2. **Helice en espace plat.** Utiliser un test unitaire separe en temps physique, sans expansion et sans parametrisation en redshift : integrer (11) et (12) avec le terme de Hubble supprime, B constant et aucune interaction. Ne pas appeler les routines cosmologiques en posant $H_0 = 0$, car leurs integrandes contiennent explicitement $1/H_0$. Verifier $\gamma = \text{constante}$ et R_L donne par (30).
3. **Redshift de l'impulsion.** Avec $B \neq 0$ mais sans synchrotron,

$$\|\mathbf{u}_f\| = \|\mathbf{u}_i\| \frac{1 + z_f}{1 + z_i}. \quad (44)$$

4. **Signe de charge.** Pour les memes conditions initiales, e^- et e^+ doivent etre deflechis en sens opposes.
5. **Axes magnetiques.** Tester successivement $\mathbf{B}_{\text{phys}} \parallel x$, $\mathbf{B}_{\text{phys}} \parallel y$, $\mathbf{B}_{\text{phys}} \parallel z$ et une direction aleatoire.
6. **Petit angle.** Pour un court trajet, verifier $\delta\theta \simeq d/R_L$ et l'ordre de grandeur de la formule de deflexion du memoire.
7. **Convergence numerique.** Diviser ϕ_{max} par deux ; l'image et les retards doivent converger.
8. **Accumulation Compton.** Comparer image et delais avec et sans accumulation. Les spectres peuvent rester proches alors que l'image change.
9. **Synchrotron desactive.** Verifier que l'activation de l'EGMF seul ne modifie pas le spectre d'energie autrement que par les effets geometriques et les conditions de detection.

12 Implementation progressive conseillee

Etape 1 - propageateur vectoriel sans synchrotron

Ajouter les routines suivantes : `EGMF_Bphys`, `rotate_rodrigues`, `propagate_one_B_cell` et `propagate_lepton_egmf`. Garder le tirage de redshift d'interaction existant. Desactiver l'accumulation Compton pour les premiers tests d'image.

Etape 2 - cellules magnetiques

Introduire une longueur de coherence et une fonction qui donne le prochain bord de cellule. Le champ peut d'abord etre uniforme, puis aleatoire par cellule. Il faut stocker les graines ou indices de cellule pour rendre les simulations reproductibles.

Etape 3 - sorties et retards

Ajouter ou verifier les colonnes de position, direction, redshift et temps. Pour un retard d'arrivee, calculer le temps cosmique de chaque emission secondaire et le trajet photonique restant jusqu'a l'observateur, puis soustraire le trajet photonique direct.

Etape 4 - synchrotron continu optionnel

Ajouter P_{syn} comme perte continue de γ . Recalculer alors le tirage d'interaction de facon incrementale, car $E_e(z)$ n'est plus analytique.

Etape 5 - emission synchrotron explicite optionnelle

Si necessaire, echantillonner le spectre synchrotron et ajouter les photons produits a la pile Monte-Carlo. Cette etape n'est pas prioritaire pour $B \sim 10^{-15}$ G.

13 Annexe A - variante si l'on veut conserver une base magnetique

La rotation de Rodrigues est recommandee. Si l'on veut tout de meme construire une base orthonormee locale, une construction robuste est :

1. $\hat{\mathbf{b}} = \mathbf{B}_{\text{phys}}/B$;
2. choisir un vecteur auxiliaire $\mathbf{a}$ non parallele a $\hat{\mathbf{b}}$, par exemple

$$\mathbf{a} = \begin{cases} (1, 0, 0) & \text{si } |b_x| < 0.9, \\ (0, 1, 0) & \text{sinon ;} \end{cases} \quad (45)$$

$$3. \hat{\mathbf{e}}_1 = [\mathbf{a} - (\mathbf{a} \cdot \hat{\mathbf{b}})\hat{\mathbf{b}}] / \|\mathbf{a} - (\mathbf{a} \cdot \hat{\mathbf{b}})\hat{\mathbf{b}}\| ;$$

$$4. \hat{\mathbf{e}}_2 = \hat{\mathbf{b}} \times \hat{\mathbf{e}}_1.$$

Les composantes associees sont alors

$$u_{\parallel} = \mathbf{u} \cdot \hat{\mathbf{b}}, \quad u_1 = \mathbf{u} \cdot \hat{\mathbf{e}}_1, \quad u_2 = \mathbf{u} \cdot \hat{\mathbf{e}}_2, \quad \theta = \text{atan2}(u_2, u_1). \quad (46)$$

Mais cette base n'apporte rien si Rodrigues est deja utilise.

14 Annexe B - liste de controle des unites

Quantite	Unite recommandee	Commentaire
c	cm s ⁻¹	$2.99792458 \times 10^{10}$
m_e	g	masse electron
e	statC	charge elementaire positive ; le signe est porte par <code>charge_sign</code>
B	G	champ physique local sauf si <code>B_mode</code> specifie comobile
H_0	s ⁻¹	convertir depuis km s ⁻¹ Mpc ⁻¹
$\mathbf{x}$	cm comobiles	avec $a_0 = 1$
$\mathbf{u}$	sans dimension	$\mathbf{p}/(m_e c)$
γ	sans dimension	$\sqrt{1 + u^2}$
ϕ	rad	signe donne par q
R_L	cm physiques	seulement pour validation locale

15 Conclusion

La formulation retenue decrit la propagation des leptons charges dans la metrique de Robertson-Walker avec une force de Lorentz et exploite le fait que le champ magnetique ne change pas l'energie lorsque le synchrotron est ignore.

La chaine de calcul conseillee est :

impulsion vectorielle $\mathbf{u} = \gamma\beta$
 $\implies$ rotation de Rodrigues autour de $\mathbf{B}_{\text{phys}}$
 $\implies$ integrale vectorielle de position

Il est plus simple a tester, plus robuste numeriquement et directement compatible avec l'objectif du memoire : calculer en trois dimensions les positions, directions, images et retards temporels des cascades.